Introduction to Proc Tabulate

GREAT LAKES SALES AND PERCENT
SUMMARY BY EMPLOYEE AND STATE

EMPLOYEE	IL		MN		WI		EMPLOYEE TOTAL	
	Sales	Pct	Sales	Pct	Sales	Pct	Sales	Pct
C AUSTIN	$10,665	29.42%	$4,056	11.19%	$21,523	59.38%	$36,244	100.00%
D CORELLI	.	.	.	.	$5,580	100.00%	$5,580	100.00%
H NUAN	.	.	$9,193	100.00%	.	.	$9,193	100.00%
M COLLDAR	.	.	.	.	$2,908	100.00%	$2,908	100.00%
M CRISTAB	.	.	$5,346	100.00%	.	.	$5,346	100.00%
T WAXWORTH	.	.	$1,279	100.00%	.	.	$1,279	100.00%
STATE TOTAL	$10,665	17.61%	$19,874	32.82%	$30,011	49.56%	$60,550	100.00%

SYSTEMS SEMINAR CONSULTANTS, INC.

Welcome to Systems Seminar Consultants, Inc.

Systems Seminar Consultants, Inc. is a SAS Alliance Quality Partner™ of SAS. Our team of SAS software experts has a broad base of knowledge and experience working with a variety of complex systems in a number of diverse industry settings. This knowledge and experience is leveraged to help you effectively achieve your business goals.

Free SAS Newsletter

Our popular publication, The Missing Semicolon™, shares SAS software solutions developed by our staff and provides additional technical assistance to our customers.

SAS Training Services

For over 1,000 students each year, we make SAS software easier to understand, use, and support.

- Public training schedules are posted on our web site.
- Private on-site training options are also available.

Additional SAS Services

SAS Consulting Services

Our staff of SAS consultants is well-versed in a variety of business areas. Our specialty areas include:

- Data systems development
- Decision support and business consultation
- Market research and analysis

SAS Help Desk Services

Our SSC team of experts is available to solve your company's daily SAS problems. Call us to develop a customized support plan that meets your company's needs.

For More Information

To receive additional information about our services or discuss a specific cost-effective solution for your company:

- Call us at: (608) 278-9964
- Check our Website at: www.sys-seminar.com
- Contact us at: 2997 Yarmouth Greenway Drive, Madison, WI 53711

Steven J. First, President

- Over 30 years of SAS experience, including hundreds of manufacturing, retail, government, marketing, and financial applications.
- Over 25 years as President and Founder of SSC
- Founder of WISAS and WISUG
- Invited speaker at local, regional, and international user groups

Course Objectives

- Identify the basic statements in a Proc Tabulate step.
- Code and run Proc Tabulate steps to analyze your data.
- Enhance the basics by using additional statements and options to:
  - Count values.
  - Compute simple statistics such as sum, mean.
  - Summarize data values at different classification levels.
  - Select a subset of rows and columns, and do accumulations.
  - Format SAS data values to make them look better.
  - Sequence your report data.

PROC TABULATE

Once our data is in shape for analysis, you may wish to display the information in a table.

PROC TABULATE is a powerful descriptive procedure that is traditionally used to create 1, 2 or 3-dimensional tables.

PROC TABULATE has a reputation as a difficult tool to learn and debug – I'll give you some tips to make it easy.

Typical One-Dimensional Reports: Columns Only

GREAT LAKES SALES AND PERCENT
SUMMARY BY CITY

CHICAGO		GRANTON		MADISON		MINNEAPOLIS		MOLINE		ST PAUL		Sales
Total	Pct	Total	Pct	Total	Pct	Total	Pct	Total	Pct	Total	Pct	Total
$6,692	15.38%	$26,038	38.46%	$3,973	7.69%	$14,528	23.07%	$3,973	7.69%	$5,346	7.69%	$60,550

GREAT LAKES SALES AND PERCENT
SUMMARY BY STATE AND CITY

IL				MN				WI				
CHICAGO		MOLINE		MINNEAPOLIS		ST PAUL		GRANTON		MADISON		Sales
Total	Pct	Total	Pct	Total	Pct	Total	Pct	Total	Pct	Total	Pct	Total
$6,692	15.38%	$3,973	7.69%	$14,528	23.07%	$5,346	7.69%	$26,038	38.46%	$3,973	7.69%	$60,550

Typical Two-Dimensional Report: Rows and Columns

GREAT LAKES SALES AND PERCENT
SUMMARY BY EMPLOYEE AND STATE

EMPLOYEE	IL		MN		WI		EMPLOYEE TOTAL	
	Sales	Pct	Sales	Pct	Sales	Pct	Sales	Pct
C AUSTIN	$10,665	29.42%	$4,056	11.19%	$21,523	59.38%	$36,244	100.00%
D CORELLI	.	.	.	.	$5,580	100.00%	$5,580	100.00%
H NUAN	.	.	$9,193	100.00%	.	.	$9,193	100.00%
M COLLDAR	.	.	.	.	$2,908	100.00%	$2,908	100.00%
M CRISTAB	.	.	$5,346	100.00%	.	.	$5,346	100.00%
T WAXWORTH	.	.	$1,279	100.00%	.	.	$1,279	100.00%
STATE TOTAL	$10,665	17.61%	$19,874	32.82%	$30,011	49.56%	$60,550	100.00%

Another Two-Dimensional Report: Rows and Columns

GREAT LAKES SALES
SUMMARY BY CUSTOMER AND EMPLOYEE

EMPLOYEE	CUSTOMER							EMPLOYEE TOTAL
	H BLAKE	J FROUDE	J GANTS	J GRANT	J GREEN	K HAMSON	P BAMSON	
C AUSTIN	$17,550	$3,973	$3,973	$6,692	$4,056	.	.	$36,244
D CORELLI	$5,580	.	.	.	.	.	.	$5,580
H NUAN	.	.	.	.	.	$9,193	.	$9,193
M COLLDAR	$2,908	.	.	.	.	.	.	$2,908
M CRISTAB	.	.	.	.	.	.	$5,346	$5,346
T WAXWORTH	.	.	.	.	$1,279	.	.	$1,279
CUSTOMER TOTAL	$26,038	$3,973	$3,973	$6,692	$5,335	$9,193	$5,346	$60,550

Another Two-Dimensional Report: Rows and Columns

GREAT LAKES SALES
SUMMARY BY STATE AND CITY

STATE	CITY	SALES
IL	CHICAGO	$6,692
	MOLINE	$3,973
MI	LANSING	$16,522
	DETROIT	$1,662
MN	MINNEAPOLIS	$14,528
	ST PAUL	$5,346
WI	GRANTON	$26,038
	MADISON	$3,973
GREAT LAKES TOTAL		$78,734

Typical Three-Dimensional Report: Pages, Rows and Columns

GREAT LAKES SALES
SUMMARY BY DATE, CUSTOMER and EMPLOYEE

DATE 01/05/08

EMPLOYEE	CUSTOMER		EMPLOYEE TOTAL
	C MACK	H BLAKE	
C AUSTIN	$1,662	$7,812	$9,474
D CORELLI	.	$5,580	$5,580
T WAXWORTH	$636	.	$636
CUSTOMER TOTAL	$2,298	$13,392	$15,690

GREAT LAKES SALES
SUMMARY BY DATE, CUSTOMER and EMPLOYEE

DATE 05/04/08

EMPLOYEE	CUSTOMER	EMPLOYEE TOTAL
	P BAMSON	
M CRISTAB	$5,346	$5,346
CUSTOMER TOTAL	$5,346	$5,346

Another Three-Dimensional Report: Pages, Rows and Columns

GREAT LAKES SALES
SUMMARY BY CUSTOMER AND EMPLOYEE

EMPLOYEE C AUSTIN

CUSTOMER	
C MACK	$1,662
H BLAKE	$17,550
J FROUDE	$3,973
J GANTS	$3,973
J GRANT	$6,692
J GREEN	$4,056
EMPLOYEE TOTAL	$37,906

GREAT LAKES SALES
SUMMARY BY CUSTOMER AND EMPLOYEE

EMPLOYEE D CORELLI

CUSTOMER	
H BLAKE	$5,580
EMPLOYEE TOTAL	$5,580

Steps For Using PROC TABULATE

1. First, make sure the data is correct.

2. Draw your table on paper. This will help you visualize which variables should be used for sub-grouping, which to use for each table dimension, and which variables to use for analysis and display.

3. Build your table in increments. Start with a simple structure and add additional features one at a time until you have arrived at the final table.

4. Concentrate on controlling the report's appearance last.

Step 1 For Using PROC TABULATE

1. **First, make sure the data in your SAS data set is correct!!!**

The SALESCUST SAS data set used in our examples contains one row for each sales transaction – identifying the customer, date, amount, etc. It was built by merging the STAFF, SALES and CUST SAS data sets.

The Sample Input SAS Dataset - SALESCUST

SALESCUST

The CONTENTS Procedure

Data Set Name	WORK.SALESCUST
Observations	45
Variables	9
Observation Length	72

Alphabetic List of Variables and Attributes

#	Variable	Type	Len	Format
7	AMOUNT	Num	8	DOLLAR7.
3	CITY	Char	18	
1	CNAME	Char	10	
2	CUSTID	Char	5	
6	DATE	Num	8	MMDDYY8.
8	ENAME	Char	10	
9	REGION	Char	2	
5	SALESID	Char	6	
4	STATE	Char	2	

The Sample Input Dataset (partial)

Salescust

SALESID	ENAME	CUSTID	CNAME	AMOUNT	REGION	STATE	CITY	DATE
900201	C AUSTIN	10001	J FROUDE	$3,973	GL	WI	MADISON	04/05/08
900386	J BILLMONT	10003	W WONDER	$2,579	NE	NY	NEW YORK	01/08/08
900386	J BILLMONT	10003	W WONDER	$8,337	NE	NY	NEW YORK	04/03/08
900386	J BILLMONT	10003	W WONDER	$6,579	NE	NY	NEW YORK	05/04/08
900009	M CRISTAB	10004	W GILBERT	$3,153	NE	MD	PENZANCE	04/06/08
900009	M CRISTAB	10004	W GILBERT	$6,724	NE	MD	PENZANCE	05/02/08
900009	M CRISTAB	10004	W GILBERT	$5,346	NE	MD	PENZANCE	05/04/08
900222	H NUAN	10004	W GILBERT	$752	NE	MD	PENZANCE	01/08/08
900009	M CRISTAB	10005	P GODOT	$7,886	S	GA	ATLANTA	01/08/08
900009	M CRISTAB	10005	P GODOT	$5,826	S	GA	ATLANTA	04/07/08
900009	M CRISTAB	10005	P GODOT	$7,028	S	GA	ATLANTA	05/03/08
900045	T WAXWORTH	10008	J GREEN	$1,279	GL	MN	MINNEAPOLIS	01/08/08
900201	C AUSTIN	10008	J GREEN	$4,056	GL	MN	MINNEAPOLIS	02/06/08

. . .

Step 2 For Using PROC TABULATE

2. Draw your table on paper.

This will help you visualize which variables should be used:

- for grouping or classification (CLASS statement)
- for each table dimension (TABLE statement)
- for analysis and display (VAR statement)

Typical Two-Dimensional Report:Rows and Columns

GREAT LAKES SALES AND PERCENT
SUMMARY BY SALES PERSON AND STATE

Sales Person	STATE		. . .		Emp. Total
	Sales	Pct	Sales	Pct	Sales
900009	$x,xxx	xx.xx%	$x,xxx	xx.xx%	$x,xxx
. . .	.	.	$xxx	xx.xx%	$x,xxx
900489	$x,xxx	xx.xx%	$x,xxx	xx.xx%	$x,xxx
State Total	$xx,xxx	xx.xx%	$xx,xxx	xx.xx%	$xx,xxx

Step 3 For Using PROC TABULATE

3. Build your table in increments.

Start with a simple structure of rows and columns. Then add additional features, such as totals and formats, one at a time, until you have the final table.

PROC TABULATE Syntax Overview

```
PROC TABULATE options;
   TABLE    [page-expression,]
            [row-expression,]
             column-expression / options;
   CLASS      variables;
   CLASSLEV*  variables/style=;
   VAR        variables;
   BY         variables;
   FREQ       variable;
   WEIGHT     variable;
   FORMAT     variables format;
   LABEL      variable='text';
   KEYLABEL   keyword ='text';
   KEYWORD*   keyword/style=;
RUN;
```

Notes:

- One TABLE statement with a VAR and/or a CLASS statement is required.
- Proc Tabulate uses: 1 SAS data set as Input and 1 printed report as Output
- *CLASSLEV and KEYWORD (ODS options, will be covered in the Customized Styles for Proc Tabulate class)

Control Statement

There are 4 basic control statements.

PROC	Names the input SAS data set for analysis
CLASS	Specifies the variables from the input SAS data set to be used for subgroups. (Like CLASS in Proc Summary/Means)
VAR	Specifies the variables from the input SAS data set to be used to generate statistics (sums, averages, etc.) (Like VAR in Proc Summary/Means.)
TABLE	Describes the report layout: 1-dimensional, 2-dimensional or 3-dimensional.

PROC Statement – Data Set Handling

Question: How do I specify which SAS data set to use and which options to use?

Answer: Use the PROC statement

- If you want to use the SAS data set Salescust, you would code:

```
PROC TABULATE DATA=SALESCUST;
```

- If you want to specify a default format for each table cell (default BEST12.2), you would code the **FORMAT=** option. It is important to set the default width large enough to accommodate all cell values.

```
PROC TABULATE DATA=SALESCUST FORMAT= 8.;
```

NOTE: Other PROC statement options will be reviewed in Step 4.

CLASS Statement - Subgrouping

Question: How do I specify the variables you want to use to subgroup my data?

Answer: Use the CLASS statement

- If you want your information summarized for each customer (CUSTID) in the SALESCUST dataset, then you would code:

  CLASS CUSTID;

- If you want information summarized for both customer and salesperson (CUSTID/ENAME) you would code:

  CLASS CUSTID ENAME;

- If you want information summarized for customer, salesperson and date (CUSTID/ENAME) you would code:

  CLASS CUSTID ENAME DATE;

Notes:

- CLASS variables can be character or numeric, normally having a few discrete values.
- CLASS variables can be grouped into fewer values with PROC FORMAT.
- Multiple CLASS statements have been allowed starting with version 8.

VAR Statement – Analysis Variables

Question: How do I specify the columns you want to use for analysis?

Answer: Use the VAR statement

- The VAR statement identifies the **numeric** variable/s you want to analyze

- If you want TABULATE to compute a statistic on the AMOUNT variable, i.e. to calculate the sum of Amount for CUSTID, you would code:

```
VAR AMOUNT;
```

Notes:

- All variables involved in math operations MUST be listed on the VAR statement.
- Multiple VAR statements have been allowed starting with version 8

TABLE Statement – Layout/Dimensions

Question: How do I specify the report layout/table dimensions in terms of column, row and page?

Answer: Use the TABLE statement - Hint: look for the commas!

- Tables can have 1, 2 or 3 dimensions composed of :

  COLUMNS

  ROWS

  PAGES

Dimensions on the TABLE statement are separated by commas:

- 1 dimensional tables specify what should appear in each **COLUMN.**
  Example: TABLE CUSTID;
- 2 dimensional tables specify what should appear in each **ROW and COLUMN.**
  Example: TABLE ENAME**,** CUSTID;
- 3 dimensional tables specify what should appear in each **PAGE, ROW and COLUMN.**
  Example: TABLE DATE**,** ENAME**,** CUSTID;

TABLE Dimensions (continued)

The following TABULATE step creates a 1 dimensional table (columns).

```
TITLE "TRANSACTION COUNT FOR GREAT LAKES STATES";
PROC TABULATE DATA=SALESCUST FORMAT=8. ;
   WHERE STATE IN ('MI','WI','MN','IL');
   CLASS STATE;
   TABLE STATE; /* NO Commas */
RUN;
```

No Commas

Notes:

- There is **NO VAR statement**. When no analysis variable is specified, each column is a count of the number of observations for each value of the CLASS variables (like PROC FREQ).

TABLE Dimensions (continued)

Since each row in the data set represents one sale, the count value is the number of sales for each state.

TRANSACTION COUNT FOR GREAT LAKES STATES

STATE			
IL	MI	MN	WI
N	N	N	N
3	6	4	6

TABLE Dimensions (continued)

A 2 dimensional table (rows and columns).

```
TITLE "TRANSACTION COUNT FOR GREAT LAKES STATES AND
  EMPLOYEE";
PROC TABULATE DATA=SALESCUST FORMAT=8. ;
   WHERE STATE IN ('MI','WI','MN','IL');
   CLASS STATE ENAME;
   TABLE STATE, ENAME; /* ONE Comma */
RUN;
```



Notes:

- Again, there is **no VAR statement** so the cells in this 2 dimensional table contain the count of the number of observations for each value of customer/salesperson combination.

TABLE Dimensions (continued)

TRANSACTION COUNT FOR GREAT LAKES STATES AND EMPLOYEE

	ENAME					
	C AUSTIN	D CORELLI	H NUAN	M COLLDAR	M CRISTAB	T WAXWORTH
	N	N	N	N	N	N
STATE						
IL	3	.	.	.	.	.
MI	1	.	2	.	2	1
MN	1	.	1	.	1	1
WI	4	1	.	1	.	.

TABLE Dimensions (continued)

A 3 dimensional table (page, row, column).

```
TITLE "TRANSACTION COUNT FOR GREAT LAKES STATES AND
  EMPLOYEE";
TITLE2 "BY DATE";
PROC TABULATE DATA=SALESCUST FORMAT=8. ;
   WHERE STATE IN ('WI','MN','IL','MI');
   FORMAT DATE MMDDYY8.;
   CLASS DATE STATE ENAME;
   TABLE DATE, STATE, ENAME; /* TWO Commas */
RUN;
```



Notes:

- Once again there is **no VAR statement** so each cell contains a count of the number of observations for each value date/customer/salesperson combination.

TABLE Dimensions (continued)

Each page in the 3 dimensional table represents a separate **date**.
Each cell shows the number of sales for each state/salesperson combination.

TRANSACTION COUNT FOR GREAT LAKES STATES AND EMPLOYEE
BY DATE

DATE 01/05/08

	ENAME		
	C AUSTIN	D CORELLI	T WAXWORTH
	N	N	N
STATE			
MI	1	0	1
WI	1	1	0

(partial output)

TABLE Dimensions (continued)

TRANSACTION COUNT FOR GREAT LAKES STATES AND EMPLOYEE
BY DATE

DATE 04/03/08

	ENAME	
	C AUSTIN	H NUAN
	N	N
STATE		
MN	0	1
WI	1	0

(partial output)

Analysis Variables – VAR Statement

Question: How do I produce statistics like sum instead of counts?

Answer: You need 2 additional things:

1. A **VAR statement** to name the variable/s you want analyzed.
2. The **crossing operator, "*".** Display values "underneath/within" a CLASS variable.

Example: Show the **sum of AMOUNT** (total dollar amount of all sales transactions) for each STATE.

```
TITLE  "GREAT LAKES SALES SUMMARY";
PROC TABULATE DATA=SALESCUST FORMAT=8. ;
   WHERE STATE IN ('MI','WI','MN','IL');
   CLASS STATE;
   VAR AMOUNT;
   TABLE STATE*AMOUNT;
RUN;
```

Crossing

Analysis Variables (continued)

GREAT LAKES SALES SUMMARY

STATE			
IL	MI	MN	WI
AMOUNT	AMOUNT	AMOUNT	AMOUNT
Sum	Sum	Sum	Sum
10665	18184	19874	30011

Notes:

- The VAR statement uses **SUM** as the default statistic (`VAR AMOUNT;`).
- The crossing operator ("*") placed the sum of amount underneath STATE (`TABLE STATE*AMOUNT*SUM;`)

Table Statistics – CROSSING Operator

Question: I want to see other summary statistics. How do I produce other statistics such as average?

Answer: Statistics are specified with the crossing operator on the TABLE statement.

```
TITLE  "GREAT LAKES SALES AVERAGE";
PROC TABULATE DATA=SALESCUST FORMAT=8. ;
   WHERE STATE IN ('MI','WI','MN','IL');
   CLASS STATE;
   VAR AMOUNT;
   TABLE STATE*MEAN*AMOUNT;
RUN;
```



Table Statistics

Now each cell contains the average sale amount for each state.

GREAT LAKES SALES AVERAGE

STATE			
IL	MI	MN	WI
Mean	Mean	Mean	Mean
AMOUNT	AMOUNT	AMOUNT	AMOUNT
3555	3031	4969	5002

Table Statistics (continued)

Some of the statistics most commonly specified on the TABLE statement:

MIN	minimum value
MAX	maximum value
RANGE	Max minus Min
SUM	sum
MEAN	averages
MODE	most frequent value
STD	standard deviation
N	number of observation with non-missing variable values
PCTSUM	percentage of sum of analysis variable values in a subgroup
PCTN	percentage of one frequency to another frequency

Others see the PROC TABULATE documentation.

Percentages

Question: Percentages are very important. Is there an easy way for me to generate percentages on my report?

Answer: Percentages can be calculated based on counts or amounts. SAS has provided some percentage statistic keywords:

• **Percentage of entire table:**	**Pctn**	**Pctsum**
• **Report percentages***	**RepPctN**	**RepPctSum**
• **Column percentages***	**ColPctN**	**ColPctSum**
• **Row percentages***	**RowPctN**	**RowPctSum**
• **Page percentages***	**PagePctN**	**PagePctSum**

Note: * New starting with Version 8.

The Crossing Operator "*"

Question: I want to see how each employee is doing in each state.

Answer: Put STATE summaries underneath EMPLOYEE values using the crossing operator.

```
TITLE "GREAT LAKES STATE SALES SUMMARY WITHIN ENAME";
PROC TABULATE DATA=SALESCUST FORMAT=8. ;
   WHERE STATE IN ('MI','WI','MN','IL');
   CLASS STATE ENAME;
   VAR AMOUNT;
   TABLE ENAME*STATE*SUM*AMOUNT;
RUN;
```



The Crossing Operator "*" (continued)

GREAT LAKES STATE SALES SUMMARY WITHIN EMPLOYEE

ENAME											
C AUSTIN				D CORELLI	H NUAN		M COLLDAR	M CRISTAB		T WAXWORTH	
STATE				STATE	STATE		STATE	STATE		STATE	
IL	MI	MN	WI	WI	MI	MN	WI	MI	MN	MI	MN
Sum	Sum	Sum	Sum	Sum	Sum	Sum	Sum	Sum	Sum	Sum	Sum
AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT
10665	1662	4056	21523	5580	12562	9193	2908	3324	5346	636	1279

Notes:

- Column headings are stacked one underneath the other as specified with the crossing operator.

Concatenation – Space Operator

Question: How do I show multiple summaries on one report?

Answer: Use the "space" operator on the TABLE statement to concatenate information within the dimension.

Example: Put ENAME summaries next to STATE summaries.

```
TITLE "GREAT LAKES SALES SUMMARY FOR STATE AND EMPLOYEE
";
 PROC TABULATE DATA=SALESCUST FORMAT=8. ;
    WHERE STATE IN ('MI','WI','MN','IL');
    CLASS STATE  ENAME;
    VAR AMOUNT;
    TABLE STATE*SUM*AMOUNT   ENAME*SUM*AMOUNT;
 RUN;
```



Notes: Concatenate rows with rows, columns with columns, pages with pages.

Concatenation Output

This output from the previous program shows the ENAME summary values next to the STATE summary values.

GREAT LAKES SALES SUMMARY FOR STATE AND EMPLOYEE

STATE				ENAME					
IL	MI	MN	WI	C AUSTIN	D CORELLI	H NUAN	M COLLDAR	M CRISTAB	T WAXWORTH
Sum	Sum	Sum	Sum	Sum	Sum	Sum	Sum	Sum	Sum
AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT
10665	18184	19874	30011	37906	5580	21755	2908	8670	1915

Multiple Table Statistics – SPACE Operator

Question: I want more than one statistic on my report. How can I display sum AND percent for example?

Answer: Statistics are specified with the Crossing AND Space operators on the TABLE statement.

```
TITLE "GREAT LAKES STATE SALES SUMMARY AND PERCENT";
PROC TABULATE DATA=SALESCUST FORMAT=8. ;
   WHERE STATE IN ('MI','WI','MN','IL');
   CLASS STATE;
   VAR AMOUNT;
   TABLE STATE*SUM*AMOUNT  STATE*PCTSUM*AMOUNT;
RUN;
```



Multiple Table Statistics

Now both totals and percents appear for each state.

GREAT LAKES STATE SALES SUMMARY AND PERCENT

STATE				STATE			
IL	MI	MN	WI	IL	MI	MN	WI
Sum	Sum	Sum	Sum	PctSum	PctSum	PctSum	PctSum
AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT
10665	18184	19874	30011	14	23	25	38

Note:

Formatting and percentages will be covered in Step 4.

Multiple Table Statistics – SPACE Operator

Question: I want to display sum and percent next to each other?

Answer: Statistics are specified with the crossing AND space operators on the TABLE statement.

```
TITLE "GREAT LAKES STATE SALES SUMMARY AND PERCENT -
  TOGETHER";
PROC TABULATE DATA=SALESCUST FORMAT=8. ;
   WHERE STATE IN ('MI','WI','MN','IL');
   CLASS STATE;
   VAR AMOUNT;
   TABLE STATE*(SUM PCTSUM)*AMOUNT;
RUN;
```



Multiple Table Statistics

Both the individual state's totals and percents appear together.

GREAT LAKES STATE SALES SUMMARY AND PERCENT - TOGETHER

STATE							
IL		MI		MN		WI	
Sum	PctSum	Sum	PctSum	Sum	PctSum	Sum	PctSum
AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT
10665	14	18184	23	19874	25	30011	38

The Special Class ALL

Question: How can I create totals for my subgroups?

Answer: ALL is a special class variable (automatically created).

- ALL means, "show me something (a statistic) across all the values of the CLASS variable in a dimension."

```
TITLE "GREAT LAKES STATES";
TITLE2 "AVERAGE SALES FOR EACH EMPLOYEE AND ALL";
PROC TABULATE DATA=SALESCUST FORMAT=8. ;
   WHERE STATE IN ('MI','WI','MN','IL');
   CLASS ENAME;
   VAR AMOUNT;
   TABLE ENAME*MEAN*AMOUNT ALL*MEAN*AMOUNT;
RUN;
```



The Special Class ALL (continued)

GREAT LAKES STATES
AVERAGE SALES FOR EACH EMPLOYEE AND ALL

ENAME						
C AUSTIN	D CORELLI	H NUAN	M COLLDAR	M CRISTAB	T WAXWORTH	All
Mean	Mean	Mean	Mean	Mean	Mean	Mean
AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT
4212	5580	7252	2908	2890	958	4144

Notes:

- In this example the information for the ALL class is concatenated (note the "space" operator) with the information on the ENAME class.

The Associative Operator "( )"

Question: It seems as if there is a lot of typing. Are there any shortcuts I can use?

Answer: The associative operator (the parenthesis) can simplify the TABLE statement.

```
TITLE "GREAT LAKES STATES";
TITLE2 "AVERAGE SALES FOR EACH EMPLOYEE AND ALL";
PROC TABULATE DATA=SALESCUST FORMAT=8. ;
   WHERE STATE IN ('MI','WI','MN','IL');
   CLASS ENAME;
   VAR AMOUNT;
   TABLE (ENAME  ALL)*MEAN*AMOUNT;
RUN;
```



Notes:

This TABLE statement creates the exact identical report to:

```
TABLE ENAME*MEAN*AMOUNT ALL*MEAN*AMOUNT;
```

Step 4 For Using PROC TABULATE

4. Concentrate on controlling the appearance last.

Controlling Table Appearance

Question: The information in my report is correct, but how do I make my report look better?

Answer: TABULATE provides appearance control options:

- **FORMAT=** Proc statement option (discussed earlier)
- **F=** TABLE statement option
- Cell labels
- **KEYLABEL** statement
- KEYWORD statement *****
- CLASSLEVEL statement *****
- **RTS** option
- **BOX** option
- **MISSTEXT** option
- **FORMCHAR** option

Note: TITLE, WHERE, FORMAT, LABEL and OPTION statements can be used in PROC TABULATE the same way you use them in other SAS PROC steps.

***** Will be covered in the Customized Styles for Proc Tabulate class

Hints For Using PROC TABULATE

```
PROC FORMAT;
  PICTURE PCTFMT LOW-HIGH='  009.99% ';

  VALUE  $PROMOCUST
    '10001'='10001'
    '10002'='10002'
    '10003'='10003'
    '10004'='10004'
    '10005'='10005';

  VALUE   SALESDIS
       0-1000 = 'NO DISCOUNT'
    1001-5000 = '3% DISCOUNT'
    5001-HIGH = '6% DISCOUNT';

RUN;
```

Note: These user-written formats are used in our examples.

A Formatting Example

Question: How can I format the cells that appear on our report?

Answer: The **F=** cell format option on the TABLE statement allows you to control the format of each cell individually.

In our example the DOLLAR10. format has been specified for the summary cells.

```
TITLE "SALES SUMMARY";
TITLE2 "BY STATE/EMPLOYEE";
PROC TABULATE DATA=SALESCUST FORMAT=12.;
   CLASS STATE ENAME;
   VAR AMOUNT;
   TABLE STATE,
         ENAME*AMOUNT*F=DOLLAR10.;
RUN;
```

NOTE: The **FORMAT=** option on the **PROC** statement specifies the default FORMAT for each table cell. It is important to set the default width large enough to accommodate all cell values.

A Formatting Example (continued)

SALES SUMMARY
BY STATE/EMPLOYEE

	ENAME						
	C AUSTIN	D CORELLI	H NUAN	J BILLMONT	M COLLDAR	M CRISTAB	T WAXWORTH
	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT
	Sum	Sum	Sum	Sum	Sum	Sum	Sum
STATE							
CO	$2,371	0	0	0	0	$6,152	$3,671
GA	0	0	0	0	0	$20,740	0
IL	$10,665	0	0	0	0	0	0
LA	0	0	0	$16,322	0	$8,789	0
MD	0	0	$752	0	0	$15,223	0
MI	$1,662	0	$12,562	0	0	$3,324	$636
MN	$4,056	0	$9,193	0	0	$5,346	$1,279
MO	0	0	0	0	0	0	$5,569
NY	0	0	0	$17,495	0	0	0
OR	0	0	0	$12,228	0	0	$3,078
TX	0	$8,242	0	0	0	0	0
UT	0	0	0	$585	$4,501	0	0
WI	$21,523	$5,580	0	0	$2,908	0	0

Formatting Percentage Statistics

Question: How can I format the percentages that appear on our report?

Answer: Create your own format and apply it to the cells containing percents.

Without formats:

```
TITLE "GREAT LAKES STATES";
TITLE2 "STATE SALES SUMMARY AND PERCENT - NO FORMATS";
PROC TABULATE DATA=SALESCUST FORMAT=8.;
   WHERE STATE IN ('MI','WI','MN','IL');
   CLASS STATE;
   VAR AMOUNT;
   TABLE STATE*SUM*AMOUNT
         STATE*PCTSUM*AMOUNT;
RUN;
```

Formatting Percentage Statistics (continued)

Both the totals and percents appear as unformatted values.

GREAT LAKES STATES
STATE SALES SUMMARY AND PERCENT - NO FORMATS

STATE				STATE			
IL	MI	MN	WI	IL	MI	MN	WI
Sum	Sum	Sum	Sum	PctSum	PctSum	PctSum	PctSum
AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT
10665	18184	19874	30011	14	23	25	38

Formatting Percentage Statistics

With formats:

```
PROC FORMAT;
   PICTURE PCTFMT LOW-HIGH='  009.99% ';
RUN;

TITLE "GREAT LAKES STATES";
TITLE2 "STATE SALES SUMMARY AND PERCENT - WITH FORMATS";
PROC TABULATE DATA=SALESCUST FORMAT=8. ;
   WHERE STATE IN ('MI','WI','MN','IL');
   CLASS STATE;
   VAR AMOUNT;
   TABLE STATE*SUM*AMOUNT*F=DOLLAR10.
         STATE*PCTSUM*AMOUNT*F=PCTFMT.;
RUN;
```

Formatting Percentage Statistics (continued)

Now both totals and percents appear with formatted values.

GREAT LAKES STATES
STATE SALES SUMMARY AND PERCENT - WITH FORMATS

STATE				STATE			
IL	MI	MN	WI	IL	MI	MN	WI
Sum	Sum	Sum	Sum	PctSum	PctSum	PctSum	PctSum
AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT
$10,665	$18,184	$19,874	$30,011	13.54%	23.09%	25.24%	38.11%

Formatting Class Variables

The FORMAT statement can be used in combination with PROC FORMAT to group values of CLASS variables.

Example: Discounts are given monthly on large orders. Our report should show each discount level on a separate page. In this three-dimensional table, a format will categorize the monthly orders by discount level.

```
PROC FORMAT;
   VALUE SALESDIS
         0-1000 = 'NO DISCOUNT'
      1001-5000 = '3% DISCOUNT'
      5001-HIGH = '6% DISCOUNT';
RUN;

DATA SALESCUST ;
     SET SALESCUST ;
     DISSALE=AMOUNT;
RUN;
```

Formatting Class Variables (continued)

```
TITLE 'DISCOUNTED SALES - Quarter 1';
PROC TABULATE DATA=SALESCUST;
   WHERE QTR(DATE) = 1;
   CLASS CNAME DATE DISSALE;
   VAR AMOUNT;

   TABLE DISSALE=' ',
         CNAME=' ' ALL='TOTAL',
         DATE=' '     *AMOUNT=' '     *SUM=' '*F=DOLLAR10.
         ALL ='GRAND'*AMOUNT='TOTAL'*SUM=' '*F=DOLLAR10.
   ;

   FORMAT DATE    WORDDATE9.;
   FORMAT DISSALE SALESDIS.;
RUN;
```

The Resulting Output

DISCOUNTED SALES - Quarter 1

NO DISCOUNT

	January	February	GRAND TOTAL
C MACK	$636	.	$636
J GRANT	.	$442	$442
W GILBERT	$752	.	$752
TOTAL	$1,388	$442	$1,830

The Resulting Output (continued)

DISCOUNTED SALES - Quarter 1

3% DISCOUNT

	January	February	March	GRAND TOTAL
C MACK	$1,662	.	$1,662	$3,324
C SMACK	.	.	$1,662	$1,662
E HANDY	$2,993	$3,078	.	$6,071
H BLAKE	.	$2,908	.	$2,908
J GREEN	$1,279	$4,056	.	$5,335
M SPRECHER	.	.	$2,356	$2,356
R BARTO	.	.	$4,231	$4,231
V VON WALD	$3,671	.	.	$3,671
W WONDER	$2,579	.	.	$2,579
TOTAL	$12,184	$10,042	$9,911	$32,137

The Resulting Output (continued)

DISCOUNTED SALES - Quarter 1
6% DISCOUNT

	January	February	March	GRAND TOTAL
H BLAKE	$13,392	.	.	$13,392
J GRANT	.	$6,250	.	$6,250
M SPRECHER	.	$12,771	$6,433	$19,204
P GODOT	$7,886	.	.	$7,886
V VON WALD	.	.	$6,152	$6,152
TOTAL	$21,278	$19,021	$12,585	$52,884

Using Variable Labels

Question: How can I change the text that appears as the column headings on our report?

Answer: You can use the LABEL statement and cell label options.

Example: Change the column headings for the AMOUNT, ENAME and STATE variables with a LABEL statement.

```
TITLE "GREAT LAKES SALES SUMMARY";
TITLE2 "BY STATE/ENAME";
PROC TABULATE DATA=SALESCUST FORMAT=8. ;
   WHERE STATE IN ('MI','WI','MN','IL');
   LABEL AMOUNT='SALES' STATE='EMPLOYEE STATE'
         ENAME='EMPLOYEE';
   CLASS STATE ENAME;
   VAR AMOUNT;
   TABLE STATE,
         ENAME*AMOUNT*F=DOLLAR10.;
RUN;
```

Table Labels (continued)

GREAT LAKES SALES SUMMARY
BY STATE/ENAME

	EMPLOYEE					
	C AUSTIN	D CORELLI	H NUAN	M COLLDAR	M CRISTAB	T WAXWORTH
	SALES	SALES	SALES	SALES	SALES	SALES
	Sum	Sum	Sum	Sum	Sum	Sum
EMPLOYEE STATE						
IL	$10,665	.	.	.	.	.
MI	$1,662	.	$12,562	.	$3,324	$636
MN	$4,056	.	$9,193	.	$5,346	$1,279
WI	$21,523	$5,580	.	$2,908	.	.

Table Labels

You can also use the cell label options to change individual headings.

Example: Label the STATE and AMOUNT **cell headings**.

```
TITLE "GREAT LAKES SALES SUMMARY";
TITLE2 "BY STATE/ENAME";
PROC TABULATE DATA=SALESCUST FORMAT=8.;
   WHERE STATE IN ('MI','WI','MN','IL');
   CLASS STATE ENAME;
   VAR AMOUNT;
   TABLE STATE ='EMPLOYEE STATE',
         ENAME ='EMPLOYEE'
          *AMOUNT='REVENUE $$'*F=DOLLAR10.;
RUN;
```

Table Labels (continued)

GREAT LAKES SALES SUMMARY
BY STATE/ENAME

	EMPLOYEE					
	C AUSTIN	D CORELLI	H NUAN	M COLLDAR	M CRISTAB	T WAXWORTH
	REVENUE $$	REVENUE $$	REVENUE $$	REVENUE $$	REVENUE $$	REVENUE $$
	Sum	Sum	Sum	Sum	Sum	Sum
EMPLOYEE STATE						
IL	$10,665	.	.	.	.	.
MI	$1,662	.	$12,562	.	$3,324	$636
MN	$4,056	.	$9,193	.	$5,346	$1,279
WI	$21,523	$5,580	.	$2,908	.	.

Labeling Statistics

Question: Can I eliminate the column headings completely on our report?

Answer: You can eliminate the statistic id cell with a cell label.

Example: Eliminate the SUM label cell altogether.

```
TITLE "GREAT LAKES SALES SUMMARY";
TITLE2 "BY STATE/ENAME";
PROC TABULATE DATA=SALESCUST FORMAT=8.;
   WHERE STATE IN ('MI','WI','MN','IL');
   CLASS STATE ENAME;
   VAR AMOUNT;
   TABLE STATE = 'EMPLOYEE STATE',
         ENAME = 'EMPLOYEE'
          *AMOUNT*SUM=''* F=DOLLAR10.;
RUN;
```



Labeling Statistics (continued)

GREAT LAKES SALES SUMMARY
BY STATE/ENAME

	EMPLOYEE					
	C AUSTIN	D CORELLI	H NUAN	M COLLDAR	M CRISTAB	T WAXWORTH
	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT	AMOUNT
EMPLOYEE STATE						
IL	$10,665	.	.	.	.	.
MI	$1,662	.	$12,562	.	$3,324	$636
MN	$4,056	.	$9,193	.	$5,346	$1,279
WI	$21,523	$5,580	.	$2,908	.	.

KEYLABEL

Question: Can I change the column headings for statistic names on our report?

Answer: **You can change the printed name of a statistic with the KEYLABEL statement. It is like the LABEL statement for a statistic name cell.**

Example: Change the wording 'SUM' to 'Sales Sum'.

```
TITLE "GREAT LAKES SALES SUMMARY";
TITLE2 "BY STATE/ENAME";
PROC TABULATE DATA=SALESCUST FORMAT=8.;
   WHERE STATE IN ('MI','WI','MN','IL');
   LABEL AMOUNT='SALES' STATE='EMPLOYEE STATE'
         ENAME='EMPLOYEE';
   CLASS STATE ENAME;
   VAR AMOUNT;
   TABLE STATE,
         ENAME*AMOUNT*F=DOLLAR10.;
   KEYLABEL SUM='Sales Sum';
RUN;
```

KEYLABEL (continued)

GREAT LAKES SALES SUMMARY
BY STATE/ENAME

	EMPLOYEE					
	C AUSTIN	D CORELLI	H NUAN	M COLLDAR	M CRISTAB	T WAXWORTH
	SALES	SALES	SALES	SALES	SALES	SALES
	Sales Sum	**Sales Sum**	**Sales Sum**	**Sales Sum**	**Sales Sum**	**Sales Sum**
EMPLOYEE STATE						
IL	$10,665	.	.	.	.	.
MI	$1,662	.	$12,562	.	$3,324	$636
MN	$4,056	.	$9,193	.	$5,346	$1,279
WI	$21,523	$5,580	.	$2,908	.	.

FLOAT

Question: A blank label displays a blank heading. Can I eliminate the blank row headings on our report?

Answer: **You can eliminate blank row headings by creating blank labels AND using the ROW=FLOAT option.**

Example: No Suppression.

```
TITLE "GREAT LAKES SALES SUMMARY BY STATE/ENAME";
TITLE2 "WITHOUT LABEL SUPPRESION and WITHOUT ROW=FLOAT
  OPTION";
PROC TABULATE DATA=SALESCUST FORMAT=8.;
   WHERE STATE IN ('MI','WI','MN','IL');
   LABEL AMOUNT='SALES' STATE='EMPLOYEE STATE'
         ENAME='EMPLOYEE';
   CLASS STATE ENAME;
   VAR AMOUNT;
   TABLE STATE*AMOUNT*F=DOLLAR8.,
         ENAME ;
RUN;
```

FLOAT – No Suppression (continued)

GREAT LAKES SALES SUMMARY BY STATE/ENAME
WITHOUT LABEL SUPPRESION and WITHOUT ROW=FLOAT OPTION

			EMPLOYEE					
			C AUSTIN	D CORELLI	H NUAN	M COLLDAR	M CRISTAB	T WAXWORTH
EMPLOYEE STATE								
IL	SALES	Sum	$10,665	.	.	.	.	.
MI	SALES	Sum	$1,662	.	$12,562	.	$3,324	$636
MN	SALES	Sum	$4,056	.	$9,193	.	$5,346	$1,279
WI	SALES	Sum	$21,523	$5,580	.	$2,908	.	.

FLOAT – Label Suppression

Example: Label Suppression.

```
TITLE "GREAT LAKES SALES SUMMARY BY STATE/ENAME";
TITLE2 "WITH LABEL SUPPRESION and WITHOUT ROW=FLOAT
  OPTION";
PROC TABULATE DATA=SALESCUST FORMAT=8.;
   WHERE STATE IN ('MI','WI','MN','IL');
   LABEL AMOUNT='SALES' STATE='EMPLOYEE STATE'
         ENAME='EMPLOYEE';
   CLASS STATE ENAME;
   VAR AMOUNT;
   TABLE STATE*AMOUNT=''*F=DOLLAR8.,
         ENAME ;
    KEYLABEL SUM=' ';
RUN;
```

FLOAT – Label Suppression (continued)

GREAT LAKES SALES SUMMARY BY STATE/ENAME
WITH LABEL SUPPRESION and WITHOUT ROW=FLOAT OPTION

			EMPLOYEE					
			C AUSTIN	D CORELLI	H NUAN	M COLLDAR	M CRISTAB	T WAXWORTH
EMPLOYEE STATE								
IL			$10,665	.	.	.	.	.
MI			$1,662	.	$12,562	.	$3,324	$636
MN			$4,056	.	$9,193	.	$5,346	$1,279
WI			$21,523	$5,580	.	$2,908	.	.

FLOAT – Label Suppression and Row=Float

Example: Label Suppression and Row=Float Option.

```
TITLE "GREAT LAKES SALES SUMMARY BY STATE/ENAME";
TITLE2 "WITH LABEL SUPPRESION and WITH ROW=FLOAT
  OPTION";
PROC TABULATE DATA=SALESCUST FORMAT=8.;
   WHERE STATE IN ('MI','WI','MN','IL');
   LABEL AMOUNT='SALES' STATE='EMPLOYEE STATE'
         ENAME='EMPLOYEE';
   CLASS STATE ENAME;
   VAR AMOUNT;
   TABLE STATE*AMOUNT=''*F=DOLLAR8.,
         ENAME
      / ROW=FLOAT;
   KEYLABEL SUM=' ';
RUN;
```

Slash - Option Separator

FLOAT – Label Suppression and Row=Float (continued)

GREAT LAKES SALES SUMMARY BY STATE/ENAME
WITH LABEL SUPPRESION and WITH ROW=FLOAT OPTION

	EMPLOYEE					
	C AUSTIN	D CORELLI	H NUAN	M COLLDAR	M CRISTAB	T WAXWORTH
EMPLOYEE STATE						
IL	$10,665	.	.	.	.	.
MI	$1,662	.	$12,562	.	$3,324	$636
MN	$4,056	.	$9,193	.	$5,346	$1,279
WI	$21,523	$5,580	.	$2,908	.	.

RTS= Option

Question: The row heading is so wide, can I make it smaller?

Answer: **The Table statement RTS= option can be used to control the size of the row headings** (default is one-fourth of the line size!).

Example: Our STATE value is taking more space than needed, so the RTS option will limit its print positions to 10.

```
TITLE "GREAT LAKES SALES SUMMARY";
TITLE2 "BY STATE/ENAME";
PROC TABULATE DATA=SALESCUST FORMAT=8.;
   WHERE STATE IN ('MI','WI','MN','IL');
   LABEL AMOUNT='SALES' STATE='EMPLOYEE STATE'
         ENAME='EMPLOYEE';
   CLASS STATE ENAME;
   VAR AMOUNT;
   TABLE STATE,
         ENAME*AMOUNT*F=DOLLAR10.
       / RTS=10;
RUN;
```

RTS = Option (continued)

GREAT LAKES SALES SUMMARY
BY STATE/ENAME

	EMPLOYEE					
	C AUSTIN	D CORELLI	H NUAN	M COLLDAR	M CRISTAB	T WAXWORTH
	SALES	SALES	SALES	SALES	SALES	SALES
	Sum	Sum	Sum	Sum	Sum	Sum
EMPLOYEE STATE						
IL	$10,665	.	.	.	.	.
MI	$1,662	.	$12,562	.	$3,324	$636
MN	$4,056	.	$9,193	.	$5,346	$1,279
WI	$21,523	$5,580	.	$2,908	.	.

BOX= Option for Two-dimensional Tables

Question: Can I place text inside the box area of the report?

Answer: Use the BOX= option on the TABLE statement.

```
.
 TITLE "GREAT LAKES SALES SUMMARY BY STATE/ENAME";
 TITLE2 'EXAMPLE: ROW,COLUMN    WITH BOX=    ';
 PROC TABULATE DATA=SALESCUST FORMAT=8.;
    WHERE STATE IN ('MI','WI','MN','IL');
    LABEL AMOUNT='SALES' STATE='EMPLOYEE STATE'
          ENAME='EMPLOYEE';
    CLASS STATE ENAME;
    VAR AMOUNT;
    TABLE STATE,
          ENAME*AMOUNT*F=DOLLAR10.
        /RTS=11 BOX='GREAT LAKES SUMMARY';
 RUN;
```

BOX= Option for Two-dimensional Tables (cont)

GREAT LAKES SALES SUMMARY BY STATE/ENAME
EXAMPLE: ROW,COLUMN WITH BOX=

GREAT LAKES SUMMARY	EMPLOYEE					
	C AUSTIN	D CORELLI	H NUAN	M COLLDAR	M CRISTAB	T WAXWORTH
	SALES	SALES	SALES	SALES	SALES	SALES
	Sum	Sum	Sum	Sum	Sum	Sum
EMPLOYEE STATE						
IL	$10,665	.	.	.	.	.
MI	$1,662	.	$12,562	.	$3,324	$636
MN	$4,056	.	$9,193	.	$5,346	$1,279
WI	$21,523	$5,580	.	$2,908	.	.

BOX= Option for Three-dimensional Tables

Question: Can I move other headings into the box area of the table?

Answer: Use the BOX= option on the TABLE statement .

Example: First generate a **three-dimensional** table **without** the BOX option.

```
TITLE 'DISCOUNTED SALES - Quarter 1';
PROC TABULATE DATA=SALESCUST FORMAT=8.;
   WHERE QTR(DATE) = 1;
   CLASS CNAME DATE DISSALE;
   VAR AMOUNT;
   TABLE DISSALE=' ',
         CNAME=' ' ALL='TOTAL',
         DATE=' '    *AMOUNT=' '    *SUM=' '*F=DOLLAR10.
         ALL ='GRAND'*AMOUNT='TOTAL'*SUM=' '*F=DOLLAR10.;
   FORMAT DATE    WORDDATE9.;
   FORMAT DISSALE SALESDIS.;
RUN;
```

Without the BOX= Option - Three-dimensional Tables

DISCOUNTED SALES - Quarter 1

NO DISCOUNT

	January	February	GRAND TOTAL
C MACK	$636	.	$636
J GRANT	.	$442	$442
W GILBERT	$752	.	$752
TOTAL	$1,388	$442	$1,830

(Continued)

BOX= _PAGE_ for Three-dimensional Tables

Place the DISSALE inside the table.

```
TITLE 'DISCOUNTED SALES - Quarter 1';
TITLE2 'EXAMPLE: PAGE,ROW,COLUMN/BOX=_PAGE_ RTS=11';

PROC TABULATE DATA=SALESCUST FORMAT=8.;
   WHERE QTR(DATE) = 1;
   CLASS CNAME DATE DISSALE;
   VAR AMOUNT;
   TABLE DISSALE=' ',
         CNAME=' ' ALL='TOTAL',
         DATE=' '    *AMOUNT=' '    *SUM=' '*F=DOLLAR8.
         ALL ='GRAND'*AMOUNT='TOTAL'*SUM=' '*F=DOLLAR8.
    /BOX=_PAGE_ RTS=11;

   FORMAT DATE    WORDDATE9.;
   FORMAT DISSALE SALESDIS.;
RUN;
```

BOX= _PAGE_ for Three-dimensional Tables (cont)

DISCOUNTED SALES - Quarter 1
EXAMPLE: PAGE,ROW,COLUMN/BOX=_PAGE_ RTS=11

NO DISCOUNT	January	February	GRAND TOTAL
C MACK	$636	.	$636
J GRANT	.	$442	$442
W GILBERT	$752	.	$752
TOTAL	$1,388	$442	$1,830

(Continued)

Global MISSING= Option

Question: Can I change the cell value of missing, '.', to something else?

Answer: Use the global MISSING= option to control the display of missing values.

Example: Display missing values as '0' instead of '.' .

```
OPTIONS MISSING='0';
PROC TABULATE DATA=SALESCUST FORMAT=8.;
   TITLE1 ' USING THE MISSING= OPTION ';
   LABEL AMOUNT='SALES' ENAME='EMPLOYEE'
         CNAME='CUSTOMER';
   CLASS CNAME ENAME;
   VAR AMOUNT;
   TABLE CNAME='',
         ENAME*AMOUNT*F=DOLLAR10.
       / RTS=15 BOX='CUSTOMER';
RUN;
```

MISSING= Option (continued)

USING THE MISSING= OPTION

CUSTOMER	EMPLOYEE						
	C AUSTIN	D CORELLI	H NUAN	J BILLMONT	M COLLDAR	M CRISTAB	T WAXWORTH
	SALES	SALES	SALES	SALES	SALES	SALES	SALES
	Sum	Sum	Sum	Sum	Sum	Sum	Sum
C HAMBRO	0	$8,242	0	0	0	0	0
C MACK	$1,662	0	$12,562	0	0	$1,662	$636
C SMACK	0	0	0	0	0	$1,662	0
C WILHELM	0	0	0	0	0	0	$5,569

(Continued)

MISSTEXT= Option

Question: Can I display text where missing values occur?

Answer: Use the TABLE statement MISSTEXT= OPTION. This is not a global option.

Example: In order to draw attention to our missing values the report will display a text value in those cells.

```
TITLE1 ' USING THE MISSTEXT= OPTION ';
PROC TABULATE DATA=SALESCUST FORMAT=8.;
   LABEL AMOUNT='SALES' ENAME='EMPLOYEE'
         CNAME='CUSTOMER';
   CLASS CNAME ENAME;
   VAR AMOUNT;
   TABLE CNAME='',
         ENAME*AMOUNT*F=DOLLAR10.
       / RTS=15 BOX='CUSTOMER' MISSTEXT='NONE';
RUN;
```

MISSTEXT= Option (continued)

USING THE MISSTEXT= OPTION

CUSTOMER	EMPLOYEE						
	C AUSTIN	D CORELLI	H NUAN	J BILLMONT	M COLLDAR	M CRISTAB	T WAXWORTH
	SALES	SALES	SALES	SALES	SALES	SALES	SALES
	Sum	Sum	Sum	Sum	Sum	Sum	Sum
C HAMBRO	NONE	$8,242	NONE	NONE	NONE	NONE	NONE
C MACK	$1,662	NONE	$12,562	NONE	NONE	$1,662	$636
C SMACK	NONE	NONE	NONE	NONE	NONE	$1,662	NONE
C WILHELM	NONE	NONE	NONE	NONE	NONE	NONE	$5,569

(Continued)

FORMCHAR Statement

Question: Can I To eliminate the "table look" of my report?

Answer: Use the FORMCHAR PROC statement option. This specifies the 11 characters used to make up the table lines.

Example: Draw solid lines for a HP laserjet printer.

```
TITLE "GREAT LAKES SALES SUMMARY BY STATE/ENAME";
PROC TABULATE DATA=SALESCUST FORMAT=8.
                        FORMCHAR='B3C4DAC2BFC3C5B4C0C1D9'X;
   WHERE STATE IN ('MI','WI','MN','IL');
   LABEL AMOUNT='SALES' STATE='EMPLOYEE STATE'
         ENAME='EMPLOYEE';
   CLASS STATE ENAME;
   VAR AMOUNT;
   TABLE STATE,
         ENAME*AMOUNT*F=DOLLAR10.
      /RTS=11 BOX='GREAT LAKES SUMMARY';
RUN;
```

FORMCHAR Statement (continued)

GREAT LAKES SALES SUMMARY BY STATE/ENAME

GREAT LAKES SUMMARY	EMPLOYEE					
	C AUSTIN	D CORELLI	H NUAN	M COLLDAR	M CRISTAB	T WAXWORTH
	SALES	SALES	SALES	SALES	SALES	SALES
	Sum	Sum	Sum	Sum	Sum	Sum
EMPLOYEE STATE						
IL	$10,665	0	0	0	0	0
MI	$1,662	0	$12,562	0	$3,324	$636
MN	$4,056	0	$9,193	0	$5,346	$1,279
WI	$21,523	$5,580	0	$2,908	0	0

FORMCHAR Option

The FORMCHAR option is a common method of eliminating all cell "drawing" characters. To eliminate the "table look" set FORMCHAR equal to a string of 11 blanks.

```
TITLE "GREAT LAKES SALES SUMMARY BY STATE/ENAME";
PROC TABULATE DATA=SALESCUST FORMAT=8.
                              FORMCHAR='           ';
   WHERE STATE IN ('MI','WI','MN','IL');
   LABEL AMOUNT='SALES' STATE='EMPLOYEE STATE'
         ENAME='EMPLOYEE';
   CLASS STATE ENAME;
   VAR AMOUNT;
   TABLE STATE,
         ENAME*AMOUNT*F=DOLLAR10.
      /RTS=11 BOX='GREAT LAKES SUMMARY';
RUN;
```

FORMCHAR Statement (continued)

GREAT LAKES SALES SUMMARY BY STATE/ENAME

GREAT LAKES SUMMARY	EMPLOYEE					
	C AUSTIN	D CORELLI	H NUAN	M COLLDAR	M CRISTAB	T WAXWORTH
	SALES	SALES	SALES	SALES	SALES	SALES
	Sum	Sum	Sum	Sum	Sum	Sum
EMPLOYEE STATE						
IL	$10,665	0	0	0	0	0
MI	$1,662	0	$12,562	0	$3,324	$636
MN	$4,056	0	$9,193	0	$5,346	$1,279
WI	$21,523	$5,580	0	$2,908	0	0

Proc Statement – Specifying Options

Question: Are there other PROC statement options that are useful?

Answer: Use options NOSEPS, ORDER and CLASSDATA.

- To customize your table **appearance**, use NOSEPS.

```
Proc Tabulate Data=SALESCUST NOSEPS STYLE=...;
```

- To specify the **order of the values** of the class variables in your table, use ORDER=.

```
Proc Tabulate Data=SALESCUST ORDER=...;
```

- To **subset** your input SAS data set based on the values in a second SAS data set, use CLASSDATA= and EXCLUSIVE.

```
Proc Tabulate Data=SALESCUST CLASSDATA=CUST EXCLUSIVE;
```

Note:

***Style will be covered in the Customized Styles for Proc Tabulate class.

NOSEPS Option

An easy way to eliminate the horizontal "drawing" characters is to use the Proc statement no separators option - NOSEPS.

```
TITLE "GREAT LAKES SALES SUMMARY BY STATE/ENAME-NOSEPS";
PROC TABULATE DATA=SALESCUST FORMAT=12. NOSEPS ;
     WHERE STATE IN ('MI','WI','MN','IL');
     LABEL AMOUNT='SALES'
           ENAME='EMPLOYEE';
     CLASS STATE ENAME;
     VAR AMOUNT;
     TABLE STATE='',
           ENAME *AMOUNT*F=DOLLAR12.2
                 / RTS=11 BOX= 'STATE';
     KEYLABEL SUM='Sum Of Sales';
RUN;
```

NOSEPS (continued)

GREAT LAKES SALES SUMMARY BY STATE/ENAME-NOSEPS

STATE	EMPLOYEE					
	C AUSTIN	D CORELLI	H NUAN	M COLLDAR	M CRISTAB	T WAXWORTH
	SALES	SALES	SALES	SALES	SALES	SALES
	Sum Of Sales	Sum Of Sales	Sum Of Sales	Sum Of Sales	Sum Of Sales	Sum Of Sales
IL	$10,665.00	0	0	0	0	0
MI	$1,662.00	0	$12,562.00	0	$3,324.00	$636.00
MN	$4,056.00	0	$9,193.00	0	$5,346.00	$1,279.00
WI	$21,523.00	$5,580.00	0	$2,908.00	0	0

Without NOSEPS Option (Continued)

Another Example:

```
TITLE  "GREAT LAKES SALES SUMMARY" ;
TITLE2 "BY STATE AND CITY - Without NOSEPS and Without
  Indent";

PROC TABULATE DATA=SALESCUST FORMAT=8.;
   WHERE STATE IN ('MI','WI','MN','IL');
   CLASS STATE CITY;
   VAR AMOUNT;
   TABLE ((STATE*CITY) ALL='GREAT LAKES SALES TOTAL') ,
         AMOUNT='SALES'*SUM=''*F=DOLLAR10.
               / RTS=24 BOX= 'STATE';
   ;
RUN;
```

Without NOSEPS (continued)

GREAT LAKES SALES SUMMARY
BY STATE AND CITY - Without NOSEPS and Without Indent

STATE		SALES
STATE	CITY	
IL	CHICAGO	$6,692
	MOLINE	$3,973
MI	DETROIT	$1,662
	LANSING	$16,522
MN	MINNEAPOLIS	$14,528
	ST PAUL	$5,346
WI	GRANTON	$26,038
	MADISON	$3,973
GREAT LAKES SALES TOTAL		$78,734

With NOSEPS Option

Another Example:

```
TITLE  "GREAT LAKES SALES SUMMARY" ;
TITLE2 "BY STATE AND CITY - WITH NOSEPS and Without
  Indent";
PROC TABULATE DATA=SALESCUST FORMAT=8. NOSEPS ;
   WHERE STATE IN ('MI','WI','MN','IL');
   CLASS STATE CITY;
   VAR AMOUNT;
   TABLE ((STATE=''*CITY) ALL='GREAT LAKES SALES TOTAL') ,
         AMOUNT='SALES'*SUM=''*F=DOLLAR10.
               / RTS=24 BOX= 'STATE';
   ;
RUN;
```

With NOSEPS (continued)

GREAT LAKES SALES SUMMARY
BY STATE AND CITY - WITH NOSEPS and Without Indent

STATE		SALES
IL	CITY	
	CHICAGO	$6,692
	MOLINE	$3,973
MI	DETROIT	$1,662
	LANSING	$16,522
MN	MINNEAPOLIS	$14,528
	ST PAUL	$5,346
WI	GRANTON	$26,038
	MADISON	$3,973
GREAT LAKES SALES TOTAL		$78,734

With NOSEPS and INDENT Options (Continued)

Another Example:

```
TITLE2 "BY STATE AND CITY - WITH NOSEPS and WITH INDENT";

PROC TABULATE DATA=SALESCUST FORMAT=8. NOSEPS;
   WHERE STATE IN ('MI','WI','MN','IL');
   CLASS STATE CITY;
   VAR AMOUNT;
   TABLE ((STATE*CITY) all='Mid-West Total'),
         AMOUNT='Sales'*SUM=''*F=DOLLAR10.
   / RTS=24  INDENT=3
   ;
RUN;
```

With NOSEPS and INDENT Options (continued)

GREAT LAKES SALES SUMMARY
BY STATE AND CITY - WITH NOSEPS and WITH INDENT

STATE	SALES
IL	
CHICAGO	$6,692
MOLINE	$3,973
MI	
DETROIT	$1,662
LANSING	$16,522
MN	
MINNEAPOLIS	$14,528
ST PAUL	$5,346
WI	
GRANTON	$26,038
MADISON	$3,973
GREAT LAKES SALES TOTAL	$78,734

PROC Statement – ORDER= Option

Question: How can I specify the order of the values of the class variables?

Answer: Use Proc statement ORDER= option.

```
TITLE 'GREAT LAKES SUMMARY - ORDER=UNFORMATTED';
PROC TABULATE DATA=SALESCUST   FORMAT=8. ORDER=UNFORMATTED;
   WHERE STATE IN ('MI','WI','MN','IL');
   CLASS DATE STATE;
   VAR AMOUNT;
   TABLE DATE ALL='State Totals',
         (STATE ALL)
          *( SUM=''*F=DOLLAR8. N=''*F=COMMA5. )
          * AMOUNT=''
   / RTS=10;
RUN;
```

Note: This is the default and is the same as Order=Internal – SAS date and alphabetical state abbreviation.

PROC Statement (continued)

GREAT LAKES SUMMARY - ORDER=UNFORMATTED

	STATE								All	
	IL		MI		MN		WI			
DATE										
01/05/08	0	0	$2,298	2	0	0	$13,392	2	$15,690	4
01/08/08	0	0	0	0	$1,279	1	0	0	$1,279	1
02/03/08	$442	1	0	0	0	0	$2,908	1	$3,350	2
02/06/08	$6,250	1	0	0	$4,056	1	0	0	$10,306	2
03/07/08	0	0	$3,324	2	0	0	0	0	$3,324	2
04/03/08	0	0	0	0	$9,193	1	$8,865	1	$18,058	2
04/05/08	$3,973	1	0	0	0	0	$3,973	1	$7,946	2
05/02/08	0	0	$3,836	1	0	0	$873	1	$4,709	2
05/03/08	0	0	$8,726	1	0	0	0	0	$8,726	1
05/04/08	0	0	0	0	$5,346	1	0	0	$5,346	1
State Totals	$10,665	3	$18,184	6	$19,874	4	$30,011	6	$78,734	19

PROC Statement – ORDER= Option

Question: How can I specify the order of the values of the class variables?

Answer: Use Proc statement ORDER= option.

```
TITLE 'GREAT LAKES SUMMARY - ORDER=DATA';
PROC TABULATE DATA=SALESCUST   FORMAT=8. ORDER=DATA;
   WHERE STATE IN ('MI','WI','MN','IL');
   CLASS DATE STATE;
   VAR AMOUNT;
   TABLE DATE ALL='State Totals',
         (STATE ALL)
          *( SUM=''*F=DOLLAR8. N=''*F=COMMA5. )
          * AMOUNT=''
   / RTS=10;
RUN;
```

Note: This is the order of the information in the input data set.

PROC Statement (continued)

GREAT LAKES SUMMARY - ORDER=DATA

	STATE								All	
	WI		MN		IL		MI			
DATE										
04/05/08	$3,973	1	0	0	$3,973	1	0	0	$7,946	2
01/08/08	0	0	$1,279	1	0	0	0	0	$1,279	1
02/06/08	0	0	$4,056	1	$6,250	1	0	0	$10,306	2
02/03/08	$2,908	1	0	0	$442	1	0	0	$3,350	2
01/05/08	$13,392	2	0	0	0	0	$2,298	2	$15,690	4
04/03/08	$8,865	1	$9,193	1	0	0	0	0	$18,058	2
05/02/08	$873	1	0	0	0	0	$3,836	1	$4,709	2
03/07/08	0	0	0	0	0	0	$3,324	2	$3,324	2
05/03/08	0	0	0	0	0	0	$8,726	1	$8,726	1
05/04/08	0	0	$5,346	1	0	0	0	0	$5,346	1
State Totals	$30,011	6	$19,874	4	$10,665	3	$18,184	6	$78,734	19

PROC Statement – ORDER= Option

Question: How can I specify the order of the values of the class variables?

Answer: Use Proc statement ORDER= option.

```
TITLE 'GREAT LAKES SUMMARY - ORDER=FREQ';
PROC TABULATE DATA=SALESCUST   FORMAT=8. ORDER=FREQ;
   WHERE STATE IN ('MI','WI','MN','IL');
   CLASS DATE STATE;
   VAR AMOUNT;
   TABLE DATE ALL='State Totals',
         (STATE ALL)
          *( SUM=''*F=DOLLAR8. N=''*F=COMMA5. )
          * AMOUNT=''
   / RTS=10;
RUN;
```

Note: This is the order based on the number of observation.

PROC Statement (continued)

GREAT LAKES SUMMARY - ORDER=FREQ

	STATE								All	
	WI		MI		MN		IL			
DATE										
01/05/08	$13,392	2	$2,298	2	0	0	0	0	$15,690	4
04/05/08	$3,973	1	0	0	0	0	$3,973	1	$7,946	2
02/06/08	0	0	0	0	$4,056	1	$6,250	1	$10,306	2
02/03/08	$2,908	1	0	0	0	0	$442	1	$3,350	2
04/03/08	$8,865	1	0	0	$9,193	1	0	0	$18,058	2
05/02/08	$873	1	$3,836	1	0	0	0	0	$4,709	2
03/07/08	0	0	$3,324	2	0	0	0	0	$3,324	2
01/08/08	0	0	0	0	$1,279	1	0	0	$1,279	1
05/03/08	0	0	$8,726	1	0	0	0	0	$8,726	1
05/04/08	0	0	0	0	$5,346	1	0	0	$5,346	1
State Totals	$30,011	6	$18,184	6	$19,874	4	$10,665	3	$78,734	19

The Sample Database – SALESCUST (partial)

THE SALESCUST SAS DATASET
The FREQ Procedure

CUSTID	Frequency
10001	**1**
10003	**3**
10004	4
10005	3
10008	2
10010	5
10012	2
10014	3
10015	1
10016	3
.....	

Note: Custid 10002 is missing from the SALESCUST data set.

Subsetting In PROC TABULATE

Question: How can I subset the information in my SAS data set?

Answer: There are several ways to subset your data set.

- Use a **WHERE** statement
- Use **PRELOADFMT** and the **PRINTMISS** options
- Use the **CLASSDATA** and **EXCLUSIVE** options

Subsetting – WHERE Statement

One way to restrict the observations on the report is by using a **WHERE** statement.

```
TITLE "SALES For Custids 10001-10005 - USING WHERE";
PROC TABULATE DATA=SALESCUST;
   LABEL AMOUNT='Sales'
         CUSTID='Customer ID';
   WHERE CUSTID between '10001' and '10005';
   CLASS CUSTID;
   VAR AMOUNT;
   TABLE CUSTID*SUM*AMOUNT*F=DOLLAR8. ;
RUN;
```

Notes:

- Since Custid 10002 is missing from the SALES data set, it does not appear in the table.

Subsetting - WHERE Statement

SALES For Custids 10001-10005 - USING WHERE

Customer ID			
10001	**10003**	10004	10005
Sum	Sum	Sum	Sum
Sales	Sales	Sales	Sales
$3,973	$17,495	$15,975	$20,740

Subsetting - PRELOADFMT and PRINTMISS Options

Question: How can I subset the information in my SAS data set?

Answer: Use PRELOADFMT and the PRINTMISS options

Example: All promotion customers should appear on the report, even if they had no sales (Customer 10002).

- Use the **PRELOADFMT** option on the CLASS statement for CUSTID – this uses the format to determine the available values.

- Use the **PRINTMISS** option on the TABLE statement to indicate that a row should appear in the report EVEN IF the values are missing.

Note: The WHERE is still needed to limit the observations/rows.
Use two CLASS statements.

Subsetting - PRELOADFMT and PRINTMISS Options

```
PROC FORMAT;
   VALUE $PROMOCUST
     '10001'='10001'
     '10002'='10002'
     '10003'='10003'
     '10004'='10004'
     '10005'='10005';
RUN;

PROC TABULATE DATA=SALESCUST;
  TITLE1 'MONTHLY SALES FOR ALL PROMOTIONAL CUSTOMERS';
  WHERE CUSTID BETWEEN '10001' AND '10005';
  LABEL AMOUNT='Sales' CUSTID='Customer ID';
  FORMAT DATE WORDDATE9. CUSTID $PROMOCUST.;
  CLASS DATE;
  CLASS CUSTID / PRELOADFMT;
  VAR AMOUNT;
  TABLE DATE,
      CUSTID='Customer ID'*SUM*AMOUNT*F=DOLLAR8.
      / RTS=10  PRINTMISS;
RUN;
```

PRELOADFMT and PRINTMISS Options (continued)

MONTHLY SALES FOR ALL PROMOTIONAL CUSTOMERS

	Customer ID				
	10001	10002	10003	10004	10005
	Sum	Sum	Sum	Sum	Sum
	Sales	Sales	Sales	Sales	Sales
DATE					
January	0	0	$2,579	$752	$7,886
April	$3,973	0	$8,337	$3,153	$5,826
May	0	0	$6,579	$12,070	$7,028

The Sample Database – PROMO_SELECT

THE PROMO_SELECT SAS DATASET

Obs	CUSTID
1	10001
2	10002
3	10003
4	10004
5	10005

Subsetting - CLASSDATA and EXCLUSIVE Options

Question: How can I subset the information in my SAS data set using a second SAS data set that contains the customers that were sent promotional materials?

Answer: In order to determine the promotion's impact on sales , ALL and ONLY those customers that received the promotion should appear on the report.

- **Use the CLASSDATA option to point to the second dataset**
- **Use the EXCLUSIVE option so that ONLY customers specified in that dataset are included**

```
PROC TABULATE DATA=SALESCUST
              CLASSDATA=PROMO_SELECT EXCLUSIVE;
 TITLE "Total Sales For Each Custid - USING CLASSDATA &
  EXCLUSIVE";
 LABEL AMOUNT='Sales'
       CUSTID='Customer ID';
 CLASS CUSTID;
 VAR AMOUNT;
 TABLE CUSTID*SUM*AMOUNT*F=DOLLAR8.;
RUN;
```

NOTE: The WHERE statement is NOT needed.

Subsetting – CLASSDATA and EXCLUSIVE Options

SALES For Custids 10001-10005 - USING CLASSDATA & EXCLUSIVE

Customer ID				
10001	10002	10003	10004	10005
Sum	Sum	Sum	Sum	Sum
Sales	Sales	Sales	Sales	Sales
$3,973	0	$17,495	$15,975	$20,740

Creating ‘Good Looking’ PROC TABULATE Reports

Considerations when generating reports using Proc Tabulate:

- Use the BOX option for labels of the first classification variable.
- Use descriptive titles and headings
- Eliminate/remove unnecessary headings and labels from the table
- For presentation and readability, consider using the variable/s with the fewest number of values for the column dimensions, and the ones with the most for the row dimensions
- Watch column widths.
- Consider displaying large sums without decimals
- Know when there is too much information for one report and create multiple reports.

Contact Us

SYSTEMS SEMINAR CONSULTANTS, INC.

SAS® Consulting, Training & Help Desk Services

2997 Yarmouth Greenway Drive • Madison, WI 53711

(608) 278-9964 x308 • Fax (608) 278-0065

www.sys-seminar.com

SAS Alliance Silver Member

Rosalind Gusinow

Senior Consultant

rgusinow@sys-seminar.com

SAS Certified Professional Version 8